> Soluzioni del compito di Calcolo Numerico del 27febbraio 2004

> Enrico Bertolazzi

```
> restart:
  with(LinearAlgebra) :
```

Esercizio 1

matrice identita e vettori canonici della base usati per i conti successivi

```
> e1,e2,e3,e4 := <1,0,0,0>,<0,1,0,0>,<0,0,1,0>,<0,0,0,1>;
```

$$e1, e2, e3, e4 := \begin{bmatrix} 1 \\ 0 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 0 \\ 1 \end{bmatrix}$$

Matrice identità

```
> ID := <e1|e2|e3|e4>;
```

$$ID := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Matrice iniziale

```
> A := Matrix([[-1,  0, -9/4, 4],
               [-1, -2, 4, -1],
               [-1,  4, -1, -2],
               [2,  0, -1, -1]]);
```

$$A := \begin{bmatrix} -1 & 0 & \frac{-9}{4} & 4 \\ -1 & -2 & 4 & -1 \\ -1 & 4 & -1 & -2 \\ 2 & 0 & -1 & -1 \end{bmatrix}$$

```
> A0 := A :
```

Matrice di permutazione

```
> P1 := <e4|e2|e3|e1>;
```

$$P1 := \begin{bmatrix} 0 & 0 & 0 & 1 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 1 & 0 & 0 & 0 \end{bmatrix}$$

Scambio la prima con la quarta riga

```
> P1A0 := P1.A0;
```

$$P1A0 := \begin{bmatrix} 2 & 0 & -1 & -1 \\ -1 & -2 & 4 & -1 \\ -1 & 4 & -1 & -2 \\ -1 & 0 & \frac{-9}{4} & 4 \end{bmatrix}$$

Matrice di eliminazione

```
> L1 := ID-<0,P1A0[2,1],P1A0[3,1],P1A0[4,1]>.Transpose(e1)/P1A0[1,1];
```

$$L1 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ \frac{1}{2} & 1 & 0 & 0 \\ \frac{1}{2} & 0 & 1 & 0 \\ \frac{1}{2} & 0 & 0 & 1 \end{bmatrix}$$

Primo passo del metodo di Gauss

```
> A1 := L1.P1A0;
```

$$A1 := \begin{bmatrix} 2 & 0 & -1 & -1 \\ 0 & -2 & \frac{7}{2} & \frac{-3}{2} \\ 0 & 4 & \frac{-3}{2} & \frac{-5}{2} \\ 0 & 0 & \frac{-11}{4} & \frac{7}{2} \end{bmatrix}$$

Matrice di scambio seconda e terza riga

```
> P2 := <e1|e3|e2|e4>;
```

$$P2 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Scambio la seconda con la terza riga

```
> P2A1 := P2.A1;
```

$$P2A1 := \begin{bmatrix} 2 & 0 & -1 & -1 \\ 0 & 4 & \frac{-3}{2} & \frac{-5}{2} \\ 0 & -2 & \frac{7}{2} & \frac{-3}{2} \\ 0 & 0 & \frac{-11}{4} & \frac{7}{2} \end{bmatrix}$$

Matrice di eliminazione

```
> L2 := ID-<0,0,P2A1[3,2],P2A1[4,2]>.Transpose(e2)/P2A1[2,2];
```

$$L2 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & \frac{1}{2} & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Secondo Passo del metodo di Gauss

```
> A2 := L2.P2A1;
```

$$A2 := \begin{bmatrix} 2 & 0 & -1 & -1 \\ 0 & 4 & \frac{-3}{2} & \frac{-5}{2} \\ 0 & 0 & \frac{11}{4} & \frac{-11}{4} \\ 0 & 0 & \frac{-11}{4} & \frac{7}{2} \end{bmatrix}$$

Nessuno scambio

```
> P3 := <e1|e2|e3|e4>;
```

$$P3 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Nessuno Scambio

```
> P3A2 := P3.A2;
```

$$P3A2 := \begin{bmatrix} 2 & 0 & -1 & -1 \\ 0 & 4 & \frac{-3}{2} & \frac{-5}{2} \\ 0 & 0 & \frac{11}{4} & \frac{-11}{4} \\ 0 & 0 & \frac{-11}{4} & \frac{7}{2} \end{bmatrix}$$

Matrice di eliminazione

```
> L3 := ID-<0,0,0,P3A2[4,3]>.Transpose(e3)/P3A2[3,3];
```

$$L3 := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 1 & 1 \end{bmatrix}$$

Terzo passo del metodo di Gauss

```
> A3 := L3.P3A2;
```

$$A3 := \begin{bmatrix} 2 & 0 & -1 & -1 \\ 0 & 4 & \frac{-3}{2} & \frac{-5}{2} \\ 0 & 0 & \frac{11}{4} & \frac{-11}{4} \\ 0 & 0 & 0 & \frac{3}{4} \end{bmatrix}$$

Matrici P, L, U

```
> P := P3.P2.P1:
  L := P.(L3.P3.L2.P2.L1.P1)^(-1):
  U := A3:
  P, L, U ;
```

$$\begin{bmatrix} 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 & 0 \\ -\frac{1}{2} & 1 & 0 & 0 \\ -\frac{1}{2} & -\frac{1}{2} & 1 & 0 \\ -\frac{1}{2} & 0 & -1 & 1 \end{bmatrix}, \begin{bmatrix} 2 & 0 & -1 & -1 \\ 0 & 4 & -\frac{3}{2} & -\frac{5}{2} \\ 0 & 0 & \frac{11}{4} & -\frac{11}{4} \\ 0 & 0 & 0 & \frac{3}{4} \end{bmatrix}$$

Controllo

```
> R := P.A - L.U;
```

$$R := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$$

Calcolo la soluzione del primo problema

```
> b  := A.< 1, 2, 2, 1>:
  Pb := P.b:
  z  := L^(-1).Pb:
  x  := U^(-1).z:
  b, Pb, z, x ;
```

$$\begin{bmatrix} -\frac{3}{2} \\ 2 \\ 3 \\ -1 \end{bmatrix}, \begin{bmatrix} -1 \\ 3 \\ 2 \\ -\frac{3}{2} \end{bmatrix}, \begin{bmatrix} -1 \\ \frac{5}{2} \\ \frac{11}{4} \\ \frac{3}{4} \end{bmatrix}, \begin{bmatrix} 1 \\ 2 \\ 2 \\ 1 \end{bmatrix}$$

Calcolo la soluzione del secondo problema

```
> b  := A.< 3, 1, -1, 0>:
  Pb := P.b:
  z  := L^(-1).Pb:
  x  := U^(-1).z:
  b, Pb, z, x ;
```

$$\begin{bmatrix} -\frac{3}{4} \\ -9 \\ 2 \\ 7 \end{bmatrix}, \begin{bmatrix} 7 \\ 2 \\ -9 \\ -\frac{3}{4} \end{bmatrix}, \begin{bmatrix} 7 \\ \frac{11}{2} \\ -\frac{11}{4} \\ 0 \end{bmatrix}, \begin{bmatrix} 3 \\ 1 \\ -1 \\ 0 \end{bmatrix}$$

Esercizio 2

```
> x := 'x' ;
```

$$x := x$$

Punti di interpolazione:

```
> X := [0,1,2,-1,-2];
  Y := [1,1,15,3,19];
```

$$X := [0, 1, 2, -1, -2]$$

$$Y := [1, 1, 15, 3, 19]$$

Polinomio interpolante:

```
> interp( X, Y, 'z');
```

$$z^4 - z + 1$$

Polinomi di Lagrange

```
> F1 := (x-X[2])*(x-X[3])*(x-X[4])*(x-X[5]);
```

$$F1 := (x - 1)\,(x - 2)\,(x + 1)\,(x + 2)$$

```
> L1 := expand(F1/subs(x=X[1],F1)) ;
```

$$L1 := \frac{1}{4}x^4 - \frac{5}{4}x^2 + 1$$

```
> F2 := (x-X[1])*(x-X[3])*(x-X[4])*(x-X[5]);
```

$$F2 := x\,(x - 2)\,(x + 1)\,(x + 2)$$

```
> L2 := expand(F2/subs(x=X[2],F2)) ;
```

$$L2 := -\frac{1}{6}x^4 - \frac{1}{6}x^3 + \frac{2}{3}x^2 + \frac{2}{3}x$$

```
> F3 := (x-X[1])*(x-X[2])*(x-X[4])*(x-X[5]);
```

$$F3 := x\,(x - 1)\,(x + 1)\,(x + 2)$$

```
> L3 := expand(F3/subs(x=X[3],F3)) ;
```

$$L3 := \frac{1}{24}x^4 + \frac{1}{12}x^3 - \frac{1}{24}x^2 - \frac{1}{12}x$$

```
> F4 := (x-X[1])*(x-X[2])*(x-X[3])*(x-X[5]);
```

$$F4 := x\,(x - 1)\,(x - 2)\,(x + 2)$$

```
> L4 := expand(F4/subs(x=X[4],F4)) ;
```

$$L4 := -\frac{1}{6}x^4 + \frac{1}{6}x^3 + \frac{2}{3}x^2 - \frac{2}{3}x$$

```
> F5 := (x-X[1])*(x-X[2])*(x-X[3])*(x-X[4]);
```

$$F5 := x\,(x - 1)\,(x - 2)\,(x + 1)$$

```
> L5 := expand(F5/subs(x=X[5],F5)) ;
```

$$L5 := \frac{1}{24}x^4 - \frac{1}{12}x^3 - \frac{1}{24}x^2 + \frac{1}{12}x$$

Polinomio interpolante

```
> p := L1*Y[1] + L2*Y[2] + L3*Y[3] + L4*Y[4] + L5*Y[5] ;
```

$$p := x^4 + 1 - x$$

Esercizio 3

```
> h := 'h';
```

$$h := h$$

Funzione da integrare

```
> f := (x,y) -> x^2 - 2*x*y ;
```

$$f := (x, y) \rightarrow x^2 - 2\,x\,y$$

Derivata prima

```
> dy := f(x,y) ;
```

$$dy := x^2 - 2\,x\,y$$

Derivata seconda

```
> ddy := expand(diff(dy,x)+diff(dy,y)*f(x,y)) ;
```

$$ddy := 2\,x - 2\,y - 2\,x^3 + 4\,x^2\,y$$

Derivata terza

```
> dddy := expand(diff(ddy,x)+diff(ddy,y)*f(x,y)) ;
```

$$dddy := 2 - 8\,x^2 + 12\,x\,y + 4\,x^4 - 8\,x^3\,y$$

Costruzione della serie di Taylor troncata

```
> ynew := unapply( y+dy*h+ddy*h^2/2+dddy*h^3/6, x, y, h) ;
```

$$ynew := (x, y, h) \rightarrow y + (x^2 - 2\,x\,y)\,h + \frac{1}{2}(2\,x - 2\,y - 2\,x^3 + 4\,x^2\,y)\,h^2 + \frac{1}{6}(2 - 8\,x^2 + 12\,x\,y + 4\,x^4 - 8\,x^3\,y)\,h^3$$

Calcolo alcuni passi

```
> x0, y0 := 1, -1;
```

$$x0, y0 := 1, -1$$

```
> h := 0.1 ;
```

$$h := 0.1$$

```
> x1 := x0 + h ;
  y1 := ynew(x0, y0, h) ;
```

$$x1 := 1.1$$

$$y1 := -.7110000000$$

```
> x2 := x1 + h ;
  y2 := ynew(x1, y1, h) ;
```

$$x2 := 1.2$$

$$y2 := -.4465925453$$

```
> x3 := x2 + h ;
  y3 := ynew(x2, y2, h) ;
```

$$x3 := 1.3$$

$$y3 := -.2093334138$$

```
>
```

Esercizio 4

```
> # Punti di interpolazione
```

```
> X0,X1,X2,X3,X4 := 0,1,2,3,4 ;
  Y0,Y1,Y2,Y3,Y4 := 1,2,9,28,65 ;
```

$$X0, X1, X2, X3, X4 := 0, 1, 2, 3, 4$$

$$Y0, Y1, Y2, Y3, Y4 := 1, 2, 9, 28, 65$$

```
> # condizioni al contorno
  S2_init  := 0 ;
  S2_final := 16 ;
```

$$S2\_init := 0$$

$$S2\_final := 16$$

```
> # Equazioni dei momenti
  # notate che h[k] = 1 costante perche i punti sono
  # equispaziati
  EQ1      := M0*(1/2) + 2*M1 + M2*(1/2) = 3*((Y2-Y1)-(Y1-Y0)) ;
  EQ2      := M1*(1/2) + 2*M2 + M3*(1/2) = 3*((Y3-Y2)-(Y2-Y1)) ;
  EQ3      := M2*(1/2) + 2*M3 + M4*(1/2) = 3*((Y4-Y3)-(Y3-Y2)) ;
  EQINIT   := M0 = S2_init ;
  EQFINAL  := M4 = S2_final ;
```

$$EQ1 := \frac{1}{2} M0 + 2\,M1 + \frac{1}{2} M2 = 18$$

$$EQ2 := \frac{1}{2} M1 + 2\,M2 + \frac{1}{2} M3 = 36$$

$$EQ3 := \frac{1}{2} M2 + 2\,M3 + \frac{1}{2} M4 = 54$$

$$EQINIT := M0 = 0$$

$$EQFINAL := M4 = 16$$

```
> RES := solve({EQ1,EQ2,EQ3,EQINIT,EQFINAL},{M0,M1,M2,M3,M4}) ;
```

$$RES := \left\{ M3 = \frac{141}{7}, M1 = \frac{43}{7}, M2 = \frac{80}{7}, M0 = 0, M4 = 16 \right\}$$

```
> # Calcolo i tratti di cubica
  S1 := Y0 + (Y1-Y0 - (M1+2*M0) / 6) * (x-X0)
          + (M0/2)*(x-X0)^2
          + ((M1-M0)/6)*(x-X0)^3 ;
  subs(RES,S1) ;
```

$$S1 := 1 + \left(1 - \frac{1}{6} M1 - \frac{1}{3} M0 \right) x + \frac{1}{2} M0\, x^2 + \frac{1}{6} (M1 - M0)\, x^3$$

$$1 - \frac{1}{42} x + \frac{43}{42} x^3$$

```
> S2 := Y1 + (Y2-Y1 - (M2+2*M1) / 6) * (x-X1)
          + (M1/2)*(x-X1)^2
          + ((M2-M1)/6)*(x-X1)^3 ;
  expand(subs(RES,S2)) ;
```

$$S2 := 2 + \left(7 - \frac{1}{6} M2 - \frac{1}{3} M1 \right) (x - 1) + \frac{1}{2} M1\, (x - 1)^2 + \frac{1}{6} (M2 - M1)\, (x - 1)^3$$

$$\frac{8}{7} - \frac{19}{42} x + \frac{3}{7} x^2 + \frac{37}{42} x^3$$

```
> S3 := Y2 + (Y3-Y2 - (M3+2*M2) / 6) * (x-X2)
          + (M2/2)*(x-X2)^2
          + ((M3-M2)/6)*(x-X2)^3 ;
  expand(subs(RES,S3)) ;
```

$$S3 := 9 + \left(19 - \frac{1}{6} M3 - \frac{1}{3} M2 \right) (x - 2) + \frac{1}{2} M2\, (x - 2)^2 + \frac{1}{6} (M3 - M2)\, (x - 2)^3$$

$$-\frac{24}{7} + \frac{269}{42} x - 3\, x^2 + \frac{61}{42} x^3$$

```
> S4 := Y3 + (Y4-Y3 - (M4+2*M3) / 6) * (x-X3)
          + (M3/2)*(x-X3)^2
          + ((M4-M3)/6)*(x-X3)^3 ;
  expand(subs(RES,S3)) ;
```

$$S4 := 28 + \left(37 - \frac{1}{6} M4 - \frac{1}{3} M3 \right) (x - 3) + \frac{1}{2} M3\, (x - 3)^2 + \frac{1}{6} (M4 - M3)\, (x - 3)^3$$

$$-\frac{24}{7} + \frac{269}{42} x - 3\, x^2 + \frac{61}{42} x^3$$

```
>
```